

```

> restart:
  with(Groebner):
  with(CodeTools):
> DIVALGO := proc(f::polynom,F::list(polynom),X::list(symbol),
  ord::symbol)
  local Q,R,P,L,i,j,LTP,sizeF,flag:

  sizeF := nops(F):
  Q := Array(1..sizeF):
  R := 0: (* Remainder Column. *)
  P := f: (* Intermediate Dividend. *)
  L := [seq(LT(F[j],X,ord),j=1..sizeF)]:

  while P<>0 do:
    flag := false:
    LTP := LT(P,X,ord):
    for i from 1 to sizeF while not flag do
      if L[i]<>0 then:
        if divide(LTP,L[i],'q') then
          Q[i] := Q[i]+q:
          P := P-(expand(q*F[i])):
          flag := true:
        fi:
      fi:
    od:
    if flag=false then:
      R := R+LTP:
      P := P-LTP:
    fi:
  od:

  return R:
end proc:
> LT := proc(f::polynom,X::list(symbol),ord::symbol)
  local lm,lc:

  lm,lc := LeadingTerm(f,ord(op(X))):
  return (lm*lc):
end proc:
> BBALGO := proc(F::list(polynom),X::list(symbol),ord::symbol)
  ::list(polynom):
  local G,GPrev,LMF,LMG,hash,size,i,j,sPol,sPolRem,reduceG:

```

```

G := F:
hash := table():
do
  GPrev := G:
  size := nops(G):
  for i from 1 to size do
    for j from (i+1) to size do
      LMF := LeadingMonomial(GPrev[i],ord(op(X))):
      LMG := LeadingMonomial(GPrev[j],ord(op(X))):
      if gcd(LMF,LMG)=1 then
        next:
      fi:
      if assigned(hash[i,j]) then
        next:
      fi:
      hash[i,j] := true:
      sPol := SPolynomial(GPrev[i],GPrev[j],ord(op(X))):
      sPolRem := DIVALGO(sPol,G,X,ord):
      if sPolRem<>0 then
        (* sPolRem := expand(sPolRem/lcoeff(sPolRem,
order=ord(op(X)))): *)
        G := [op(G),sPolRem]:
      fi:
    od:
  od:
until G=GPrev:
reduceG := REDUCEGB(GPrev,X,ord):
return reduceG:
end proc:

```

```

> REDUCEGB := proc(G::list(polynom),X::list(symbol),ord::symbol)
::list(polynom):
local temp,i,j,LTF,LTFtemp,LTGtemp,LTG,k,newGB:

temp := G:

i := 1:
while i<= nops(temp) do:
  LTF := LT(temp[i],X,ord):
  j := i+1:
  while j<= nops(temp) do:
    LTG := LT(temp[j],X,ord):
    if divide(LTG,LTF) then:

```

```

        temp := subsop(j=NULL,temp):
    else:
        j++:
    fi:
od:
i++:
od:

for i to nops(temp) do:
    temp[i] := DIVALGO(temp[i],subsop(i=NULL,temp),X,ord):
    if temp[i]<>0 then:
        newGB[i] := expand(temp[i]/lcoeff(temp[i])):
    fi:
od:

newGB := convert(newGB,list):

return newGB:
end proc:

```

```

> ide[1] := [x+y+z-1,x-y-(2*z),(2*x)+(2*z)-2]:
ide[2] := [(x*y)-1,(x*z)-1,(y*z)-1]:
ide[3] := [(x*y+z)-(x*z),x^2-z,(2*x^3)-(x^2*y*z)-1]:
ide[4] := [(4*x^2)+(x*y^2)-z+(1/4),(2*x)+(y^2*z)+(1/2),(x^2*z)-
(1/2)*x)-y^2]:

```

```

> IDEALS := [ide[1],ide[2],ide[3],ide[4]]:
> IDEALS;

```

```

[ [x+y+z-1, x-y-2z, 2x+2z-2], [xy-1, xz-1, yz-1], [xy-xz+z, x^2-z,
-x^2yz+2x^3-1], [4x^2+xy^2-z+1/4, 2x+y^2z+1/2, x^2z-1/2x-y^2] ]

```

```

> PLEXGB := proc(F::list(list(polynom)))
local compGB,mapGB,i:

compGB := table():
mapGB := table():
for i from 1 to nops(F) do:
    compGB[i] := BBALGO(F[i],[x,y,z],plex):
    mapGB[i] := Basis(F[i],plex(x,y,z)):
od:

```

```

compGB := convert(compGB,list):
mapGB := convert(mapGB,list):

```

```

return (compGB,mapGB):
end proc:

```

```

> r1,r2 := PLEXGB(IDEALS):
r1;
r2;

```

$$\begin{aligned}
 & \left[\left[x - \frac{2}{3}, y, z - \frac{1}{3} \right], \left[y - z, x - z, z^2 - 1 \right], \left[x + \frac{4}{7} z^5 - \frac{5}{7} z^4 - \frac{13}{7} z^3 - \frac{10}{7} z^2 - \frac{1}{7} z \right. \right. \\
 & \quad \left. \left. + \frac{2}{7}, \frac{5}{7} z^4 + \frac{13}{7} z^3 + \frac{10}{7} z^2 + y - \frac{6}{7} z - \frac{2}{7} - \frac{4}{7} z^5, z^6 - z^5 - 4z^4 - 2z^3 + 1 \right], \left[x \right. \right. \\
 & \quad \left. \left. + \frac{4638}{497} z^6 - \frac{75}{497} z^5 + \frac{2111}{3976} z^4 + \frac{61031}{1988} z^3 + \frac{232833}{3976} z^2 - \frac{85042}{497} z \right. \right. \\
 & \quad \left. \left. + \frac{144407}{1988}, -\frac{297833}{994} + \frac{1407741}{1988} z + y^2 - \frac{481837}{1988} z^2 - \frac{251555}{1988} z^3 - \frac{4197}{1988} z^4 \right. \right. \\
 & \quad \left. \left. + \frac{318}{497} z^5 - \frac{19188}{497} z^6, \frac{133}{8} z - \frac{15}{4} - \frac{171}{8} z^2 + z^7 + \frac{1}{16} z^5 + \frac{13}{4} z^4 - \frac{1}{2} z^6 \right. \right. \\
 & \quad \left. \left. + \frac{75}{16} z^3 \right] \right] \\
 & \left[[3z - 1, y, 3x - 2], [z^2 - 1, y - z, x - z], [z^6 - z^5 - 4z^4 - 2z^3 + 1, -4z^5 + 5z^4 + 13z^3 \right. \right. \\
 & \quad \left. \left. + 10z^2 + 7y - 6z - 2, 4z^5 - 5z^4 - 13z^3 - 10z^2 + 7x - z + 2 \right], [16z^7 - 8z^6 + z^5 \right. \right. \\
 & \quad \left. \left. + 52z^4 + 75z^3 - 342z^2 + 266z - 60, -76752z^6 + 1272z^5 - 4197z^4 - 251555z^3 \right. \right. \\
 & \quad \left. \left. + 1988y^2 - 481837z^2 + 1407741z - 595666, 37104z^6 - 600z^5 + 2111z^4 + 122062z^3 \right. \right. \\
 & \quad \left. \left. + 232833z^2 + 3976x - 680336z + 288814 \right] \right]
 \end{aligned}$$

```

> GRLEXGB := proc(F::list(list(polynom)))
local compGB,mapGB,i:

compGB := table():
mapGB := table():
for i from 1 to nops(F) do:
    compGB[i] := BBALGO(F[i],[x,y,z],grlex):
    mapGB[i] := Basis(F[i],grlex(x,y,z)):
od:

compGB := convert(compGB,list):
mapGB := convert(mapGB,list):

```

```
return (compGB,mapGB):
```

```
end proc:
```

```
> r1,r2 := GRLEXGB(IDEALS):
```

```
  r1;
```

```
  r2;
```

```
[ [ [  $x - \frac{2}{3}, y, z - \frac{1}{3}$  ], [  $y - z, x - z, z^2 - 1$  ], [  $yz^2 + 2yz - 2z^2 + 1, -\frac{1}{4}z^4 + \frac{3}{4}z^3 + yz$   
-  $\frac{1}{2}z^2 + \frac{1}{4}y - \frac{1}{2}z + \frac{1}{2}, x + y - z, y^2 - 2yz + z^2 - z$  ], [  $xy^2 + 8y^2 - 2z^2 + 3x$   
-  $\frac{1}{2}z + \frac{1}{4}, x + \frac{1}{2}y^2z + \frac{1}{4}, \frac{1}{2}z^2 - \frac{1}{8}z + x^2 - \frac{3}{4}x - 2y^2, -y^4 + z^2x - \frac{1}{4}xz$   
+  $32y^2 - 8z^2 + 12x - \frac{5}{2}z + \frac{9}{8}, -\frac{4}{3}y^2 - 6x - \frac{2}{3}z^3 + \frac{1}{6}z^2 + xz - \frac{4}{3}$  ] ] ]  
[ [  $3z - 1, y, 3x - 2$  ], [  $y - z, x - z, z^2 - 1$  ], [  $x + y - z, y^2 - 2yz + z^2 - z, yz^2 + 2yz - 2z^2$   
+  $1, z^4 - 3z^3 - 4yz + 2z^2 - y + 2z - 2$  ], [  $8x^2 - 16y^2 + 4z^2 - 6x - z, 4z^3 - 6xz + 8y^2$   
-  $z^2 + 36x + 8, 2y^2z + 4x + 1, 4xy^2 + 32y^2 - 8z^2 + 12x - 2z + 1, 8y^4 - 8xz^2 + 2xz$   
-  $256y^2 + 64z^2 - 96x + 20z - 9$  ] ] ]
```